

APPLE II SERIES

StripWare™

SECOND GIANT BOOK OF COMPUTER GAMES

Tim Hartnell

Contains 6 programs:

- The Big Maze
- Cavalier 'Chute
 - Engulf
- Mine Sweeper
- It's a Mad, Mad World
- Celtic Tarot



Package is not returnable if plastic wrap is cut or removed.

TIM HARTNELL, author of such current bestselling computer books as *Creating Adventure Games*, is an Australian journalist who enjoys writing on any computer-related topic. He has written more than 30 books on personal computers, including many game books.

Reprinted with permission of
Tim Hartnell and Ballantine Books

Apple and Applesoft are registered trademarks of
Apple Computer, Inc.

StripWare and Softstrip are trademarks of
Cauzin Systems, Inc.

APPLE II SERIES

StripWare™

SECOND GIANT BOOK OF COMPUTER GAMES

Tim Hartnell



Contains 6 programs:

- The Big Maze
- Cavalier 'Chute
 - Engulf
- Mine Sweeper
- It's a Mad, Mad World
- Celtic Tarot



Package is not returnable if plastic wrap is cut or removed.

Softstrip™
COMPUTER READABLE PRINT

Suggested Retail Price

\$12.98

- **Get out of tight places in
THE BIG MAZE**
- **Survive with subtle precision in
CAVALIER 'CHUTE**
- **ENGULF** aliens in far away galaxies
- **Explosive action abounds in
MINE SWEEPER**
- **IT'S A MAD, MAD WORLD** full
of adventure
- **THE CELTIC TAROT** forsees what
lies ahead

Cauzin 1994

CAUZIN SYSTEMS, INC.

835 SOUTH MAIN STREET WATERBURY, CT 06706

Welcome to The Cauzin Family

Welcome to the world of data strips. You now own a piece of a revolutionary technological advancement that bridges the gap between print and electronic media more quickly and easily than ever before.

Each data strip package contains selected highlights from among the best programs each publisher has to offer from articles, books, and software.

We at Cauzin Systems, Inc., hope you'll enjoy this software package, and believe you'll find data strips to be very practical and useful.

Your comments and ideas on this package or any additional applications you'd like to suggest, would be greatly appreciated. Please send your comments or remarks to:

Cauzin Publishing
835 South Main Street
Waterbury, Conn. 06706

© 1985



SECOND GIANT BOOK OF COMPUTER GAMES

by Tim Hartnell

THE BIG MAZE

In the world of video and computer software, Pac-Man probably did more to resurrect the concept of maze games than any other variation on the theme. However, you don't need ghosts and power pellets to have a challenging and enjoyable time in trying to make it through a maze.


THE BIG MAZE is a remarkable program from London programmer Tony Pearson. Your mission is to move through a maze (on screen you're represented by the letter I) and get to the exit (represented by X). You can request a view of the maze from above at any time. The program will draw a map showing you where you are, where you've been, and where you want to go.



The real challenge of the maze comes into play when, instead of looking at the entire maze from above, you stay "within" it, and look (or move) in a particular direction. You'll see a view of the maze for a short distance in the direction you requested. It may take a few runs of the program before you are able to visualize the maze, but it's well worth the trouble.

When you start, you'll see the commands to use in entering your moves. Enter N, S, E, or W to look in that direction. Follow it, next turn, with M to move in that direction. Enter V to view the entire maze from above, or R to move randomly to a new position. Exit at any time by pressing CONTROL-RESET.






Reading THE BIG MAZE

The following data strip contains the program described in this article. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, run the program from the Cauzin menu or enter `RUN BIG.MAZE`. Operating instructions are found in the article. Exit anytime by pressing `CONTROL-RESET`.



THE BIG MAZE
by Tony Pearson
Second Giant Book of Computer Games
Ballantine Books
Copyright 1985. All rights reserved.

THE BIG MAZE

```

10 REM THE BIG MAZE
11 REM
12 REM FROM TIM HARTNELL'S
13 REM SECOND GIANT BOOK
14 REM OF COMPUTER GAMES
15 REM
16 REM BALLANTINE BOOKS
17 REM
18 REM *****
19 REM
20 HOME
40 DIM A$(30)
50 D = 1
60 A$(5) = " *****
    *****: REM 20 *'S
70 FOR F = 6 TO 24 STEP 2
80 A$(F) = " *
    **
90 A$(F + 1) = " ** ** **
    ** * **
100 NEXT F
110 A$(25) = A$(5)
120 FOR F = 1 TO 70
130 N = INT ( RND (1) * 20) + 6
140 N1 = INT ( RND (1) * 20) + 6

150 A$(N) = MID$( A$(N),1,N1 - 1
    ) + "*" + MID$( A$(N),N1 +
    1,25 - N1)
160 NEXT F
170 N1 = INT ( RND (1) * 2) + 1
180 ON N1 GOSUB 1090,1130
190 X = INT ( RND (1) * 4) + 9
200 Y = INT ( RND (1) * 4) + 9
210 IF MID$( A$(X),Y,1) = "*"
    THEN 190
220 A$(X) = MID$( A$(X),1,Y - 1)
    + "I" + MID$( A$(X),Y + 1,
    25 - Y)
230 MO = 0
240 PRINT "ENTER 'N', 'S', 'E' O
    R 'W' TO LOOK IN"
250 PRINT "THAT DIRECTION. FOLL
    OW IT, NEXT MOVE,"
260 PRINT "WITH 'M' TO MOVE IN T
    HAT DIRECTION..."
270 PRINT "ENTER 'V' TO VIEW THE
    MAP, OR"
280 PRINT "'R' TO MOVE RANDOMLY
    TO A NEW POSITION"
290 PRINT "CHOOSE: N, S, E, W, V
    , M, R ";: GET R$: PRINT R$
300 IF R$ = "V" THEN D = 5
310 IF R$ = "R" THEN D = 6
320 IF R$ = "M" THEN D = 0
330 IF R$ = "N" THEN D = 4
340 IF R$ = "E" THEN D = 1
350 IF R$ = "W" THEN D = 3
360 IF R$ = "S" THEN D = 2
370 IF D > 0 THEN D1 = D
380 IF D = 0 THEN 530
390 IF D = 5 THEN GOSUB 680
400 IF D = 6 THEN GOSUB 730
410 IF D > 4 OR D < 1 THEN 290
420 GOSUB 1160
430 PRINT "DIRECTION: ";U$
440 M$ = "":L$ = "":R$ = ""
450 ON D1 GOSUB 830,880,950,1020

460 FOR F = 5 TO 1 STEP - 1
470 PRINT MID$( L$,F,1);
480 PRINT MID$( M$,F,1);
490 PRINT MID$( R$,F,1)
500 NEXT F
510 GOTO 290
520 REM *****
530 X1 = X:Y1 = Y
540 X = X + (D1 = 2) - (D1 = 4)
550 Y = Y - (D1 = 3) + (D1 = 1)
560 IF MID$( A$(X),Y,1) = "*" THEN
    640
570 IF MID$( A$(X),Y,1) = "X" THEN
    800
580 MO = MO + 1
590 A$(X) = MID$( A$(X),1,Y - 1)
    + "I" + MID$( A$(X),Y + 1,
    25 - Y)
600 A$(X1) = MID$( A$(X1),1,Y1 -
    1) + "." + MID$( A$(X1),Y1 +
    1,25 - Y1)
610 GOSUB 1160
620 PRINT "DIRECTION: ";U$
630 GOTO 440
640 X = X1:Y = Y1
650 PRINT "BAD MOVE"
660 GOTO 290
670 REM *****
680 HOME : FOR F = 4 TO 26
690 PRINT A$(F)
700 NEXT F
710 RETURN
720 REM *****
730 A$(X) = MID$( A$(X),1,Y - 1)
    + " " + MID$( A$(X),Y + 1,
    25 - Y)
740 X = INT ( RND (1) * 4) + 9
750 Y = INT ( RND (1) * 4) + 9
760 IF MID$( A$(X),Y,1) = "*" THEN
    730
770 A$(X) = MID$( A$(X),1,Y - 1)
    + "I" + MID$( A$(X),Y + 1,
    25 - Y)

780 RETURN
790 REM *****
800 PRINT SPC( 4);"YOU'RE OUT A
    FTER "MO" MOVES..."
810 END
820 REM *****
830 M$ = MID$( A$(X),Y,5)
840 L$ = MID$( A$(X - 1),Y,5)
850 R$ = MID$( A$(X + 1),Y,5)
860 RETURN
870 REM *****
880 FOR F = X TO X + 5
890 M$ = M$ + MID$( A$(F),Y,1)
900 R$ = R$ + MID$( A$(F),Y - 1,
    1)
910 L$ = L$ + MID$( A$(F),Y + 1,
    1)
920 NEXT F
930 RETURN
940 REM *****
950 FOR F = 0 TO - 5 STEP - 1
960 M$ = M$ + MID$( A$(X),Y + F,
    1)
970 R$ = R$ + MID$( A$(X - 1),Y +
    F,1)
980 L$ = L$ + MID$( A$(X + 1),Y +
    F,1)
990 NEXT F
1000 RETURN
1010 REM *****
1020 FOR F = X TO X - 5 STEP -
    1
1030 M$ = M$ + MID$( A$(F),Y,1)
1040 R$ = R$ + MID$( A$(F),Y + 1
    ,1)
1050 L$ = L$ + MID$( A$(F),Y - 1
    ,1)
1060 NEXT F
1070 RETURN
1080 REM *****
1090 N = INT ( RND (1) * 10) + 1
    1
1100 A$(N) = " X " + MID$(
    A$(N),12,14): REM FIVE SPAC
    ES EACH SIDE OF X
1110 RETURN
1120 REM *****
1130 N = INT ( RND (1) * 10) + 1
    1
1140 A$(N) = MID$( A$(N),1,19) +
    " X": REM FIVE SPACES
1150 RETURN
1160 IF D1 = 1 OR D = 1 THEN U$ =
    "EAST"
1170 IF D1 = 2 OR D = 2 THEN U$ =
    "SOUTH"
1180 IF D1 = 3 OR D = 3 THEN U$ =
    "WEST"
1190 IF D1 = 4 OR D = 4 THEN U$ =
    "NORTH"
1200 RETURN

```

CAVALIER 'CHUTE

Some games call for rapid-fire action while you try to destroy everything in sight. Others have more subtle play where precision timing is the necessary skill in order to master the game.

Written by Neal Cavalier-Smith, CAVALIER 'CHUTE is a program that offers the challenge of battling wayward wind-drifts while you attempt to land your parachute safely. Your target is a tiny landing-pad that is floating on the waves of a great ocean.




There are four different levels of play and the game will start after you have selected the degree of difficulty you're ready to take on. On screen you will see:

SELECT YOUR GAME:

- A — EASY
- B — MODERATE
- C — DIFFICULT
- D — INCREDIBLY HARD


The action begins with the computer printing out your parachute over and over again as it descends. In order to guide it onto the landing pad below, use the Z key to move the parachute to the left and the M key to move to the right. Any other key keeps you heading straight down. Miss the pad, and you'll drown! To exit, press CONTROL-RESET at any time.



Reading CAVALIER 'CHUTE

The following data strip contains the program described in this article. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, run the program from the Cauzin menu or enter RUN CAVALIER.CHUTE. Operating instructions are found in the article. Exit anytime by pressing CONTROL-RESET.



CAVALIER 'CHUTE
by Neal Cavalier-Smith
Second Giant Book of Computer Games
Ballantine Books
Copyright 1985. All rights reserved.



CAVALIER 'CHUTE

```


10 REM CAVALIER 'CHUTE
11 REM
12 REM FROM TIM HARTNELL'S
13 REM SECOND GIANT BOOK
14 REM OF COMPUTER GAMES
15 REM
16 REM BALLANTINE BOOKS
17 REM
18 REM *****
19 REM
20 P = 20
30 GOSUB 110: REM INITIALIZE
40 REM *****
50 REM MAIN CYCLE
60 GOSUB 360: REM PRINT 'CHUTE
70 GOSUB 520: REM LANDING STRIP
80 GOSUB 580: REM GET KEYS
90 GOTO 60
100 REM *****
110 REM INITIALIZATION
120 HOME
140 PRINT : PRINT "SELECT YOUR G
AME:"
150 PRINT SPC( 6); "A - EASY"
160 PRINT SPC( 6); "B - MODERATE
"
170 PRINT SPC( 6); "C - DIFFICUL
T"
180 PRINT SPC( 6); "D - INCREDIB
LY HARD"
185 PRINT : PRINT "USE Z AND M T
O MOVE THE CHUTE": PRINT "AN
Y OTHER KEY FOR STRAIGHT DOW
N": PRINT
187 PRINT "BE SURE CAPS LOCK IS
ON.": PRINT
190 PRINT : HTAB 20: GET R$: PRINT
R$
200 IF R$ < "A" OR R$ > "D" THEN
190
210 DF = 1 / 2
220 IF R$ = "C" THEN DF = 1
230 IF R$ = "B" THEN DF = 2
240 IF R$ = "A" THEN DF = 3
250 STP = 0: REM SCREEN TOP
280 SBOT = 19: REM SCREEN BOTTOM
290 SWDE = 40: REM WIDTH OF SCREE
N
300 SWDE = SWDE - 3
310 ACC = INT ( RND (1) * (SWDE -
4))
320 PAD = INT ( RND (1) * (SWDE -
10))
330 PLUS = 1
340 RETURN
350 REM *****
360 REM PRINT PARACHUTE
370 HOME
380 FOR C = 0 TO STP
390 PRINT
400 NEXT C
410 PRINT SPC( ACC); " ^ "
420 PRINT SPC( ACC); "^^^"
430 PRINT SPC( ACC); " O "
440 PRINT SPC( ACC); " Y "
450 IF STP = SBOT - 4 THEN 490
460 FOR C = 5 TO SBOT - STP
470 PRINT
480 NEXT C
490 STP = STP + 1
500 RETURN
510 REM *****
520 REM LANDING STRIP
530 PAD = PAD + PLUS
540 IF PAD > SWDE - 6 THEN PLUS =
- 1
550 PRINT SPC( PAD); " @@@@ "
560 RETURN
570 REM *****
580 REM GET KEYS
590 IF STP = SBOT - 3 THEN GOSUB
650
600 GET Y$
610 IF Y$ = "Z" THEN ACC = ACC -
DF
620 IF Y$ = "M" THEN ACC = ACC +
DF
625 IF ACC < 0 THEN ACC = 36
626 IF ACC > 36 THEN ACC = 0
630 RETURN
640 REM *****
650 REM SEE IF DROWNED
660 IF ACC < PAD - 1 THEN 770
670 IF ACC > PAD + 2 THEN 770
680 IF ACC = PAD + 1 OR ACC = PA
D THEN 730
690 PRINT SPC( 8); "YOU MADE IT.
..BY THE"
700 PRINT SPC( 9); "SKIN OF YOUR
TEETH"
710 P = P + 5
720 GOTO 800
730 PRINT SPC( 12); "CONGRATULAT
IONS!!"
740 PRINT SPC( 13); "A SAFE LAND
ING!"
750 P = INT ( P + 10 / DF)
760 GOTO 800
770 PRINT SPC( 12); "SPLOSSSSSHH
HHHHHHH!!"
780 PRINT SPC( 14); "YOU GOT SOA
KED!"
790 P = P - 2
800 A = A + 1
810 IF A = 20 OR P < 0 THEN 890
820 PRINT SPC( 11); "YOU HAVE "P
" POINTS"
830 FOR X = 1 TO 1000: NEXT X
850 GOSUB 250
860 RETURN
870 REM *****
880 REM DROWNED!
890 PRINT SPC( 13); "YOU'VE DROW
NED!"
900 PRINT : PRINT SPC( 12); "YOU
SCORED "P" POINTS"
910 PRINT SPC( 16); "IN "A" ATTE
MPTS."
920 END

```

ENGULF

Space, the final frontier where so much has happened over the years. Many of us confronted onrushing Invaders bent on destruction and Asteroids that not only had to be avoided but also blown apart. Well, just when you thought it might be safe to travel through the galaxy, another nasty alien stands in your way.

At the controls of your space-cruiser Borealis, you're off to a corner of the universe. Your task is to "engulf" the alien, immobilizing it by destroying all the sectors of space around it.



Both you and the alien (shown on your screen as an A) are unable to move onto the outer border of the area of space under view. These "out of bounds" areas are indicated by "-" marks (hyphens).

The alien doesn't have to change locations on every move. However, if it does decide to move, it may be by one or two squares from its present position. You must isolate your crafty foe by blasting areas of space out of existence. All you have to do is enter the coordinates of the space you want to destroy as two numbers. Press CONTROL-RESET to exit the program any time.


Wipe out all the squares onto which the alien could move and you've "engulfed" it. The alien keeps a watch on the space around it, and your computer can detect the degree of danger the alien senses in its present position, so you can follow (more or less) what the alien is thinking. A word of caution: You're not allowed to land on the alien, and you'll destroy yourself if you do.

You'll discover that the best strategy is to trap the alien against one of the sides of the grid. This limits its potential moves. Of course, the alien is aware of the danger of being caught near the sides, and uses its limited intelligence to try to keep away from them. A hint: Try to build a "fence" of blank areas in a curve around the alien so that you can force it to move into a side. Good luck!

Reading ENGULF

The following data strip contains the program described in this article. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, run the program from the Cauzin menu or enter RUN ENGULF. Operating instructions are found in the article. Exit anytime by pressing CONTROL-RESET.



ENGULF

by Tim Hartnell

Second Giant Book of Computer Games

Ballantine Books

Copyright 1985. All rights reserved.

ENGULF

```

10 REM ENGULF
11 REM
12 REM FROM TIM HARTNELL'S
13 REM SECOND GIANT BOOK
14 REM OF COMPUTER GAMES
15 REM
16 REM BALLANTINE BOOKS
17 REM
18 REM *****
19 REM
20 HSCRE = 0
30 DIM A(10,10)
40 GOSUB 1030: REM INITIALIZE
50 GOSUB 220: REM ALIEN MOVE
60 REM *****
70 REM MAIN LOOP
80 GOSUB 800: REM PRINT OUT
90 GOSUB 670: REM PLAYER MOVE
100 GOSUB 800
110 GOSUB 220: REM ALIEN MOVE
120 TME = TME - 1
130 SHOTS = SHOTS + 1
140 IF TME = 0 THEN 580
150 GOTO 80
160 REM *****
170 REM COLLISION
180 PRINT SPC( 7);"YOU HIT AN A
LIEN, CAPTAIN"
190 PRINT SPC( 7);"AND HAVE BEE
N DESTROYED"
200 GOTO 580
210 REM *****
220 REM ALIEN MOVE
230 REM * CHECK IF SURROUNDED *
240 H = 0
250 H = H + A(M - 1,N)
260 H = H + A(M + 1,N)
270 H = H + A(M,N - 1)
280 H = H + A(M,N + 1)
290 H = H + A(M - 1,N + 1)
300 H = H + A(M + 1,N + 1)
310 H = H + A(M + 1,N - 1)
320 H = H + A(M - 1,N - 1)
330 IF H = 16 THEN 500: REM SURR
OUNDED
340 REM * MOVE ALIEN *
350 E = M:F = N
360 CT = 0
370 M = M - INT ( RND (1) * 3) +
INT ( RND (1) * 3)
380 IF M < 2 OR M > 9 THEN M = E
: GOTO 370
390 IF (M < 4 OR M > 7) AND RND
(1) > .7 THEN M = E: GOTO 37
0
400 CT = CT + 1: IF CT = 60 THEN
M = E:N = F: GOTO 460
410 N = N - INT ( RND (1) * 3) +
INT ( RND (1) * 3)
420 IF N < 2 OR N > 9 THEN N = F
: GOTO 410
430 IF (N < 4 OR N > 7) AND RND
(1) > .7 THEN N = F: GOTO 41
0
440 CT = CT + 1: IF CT = 60 THEN
M = E:N = F: GOTO 460
450 IF A(M,N) = 2 THEN M = E:N =
F: GOTO 370
460 A(E,F) = 0
470 A(M,N) = 1
480 RETURN
490 REM *****
500 REM SURROUND
510 GOSUB 800
520 PRINT "-----> ENGULFED! W
ELL DONE..."
530 PRINT "-----> IT TOOK YOU
"SHOTS" SHOTS"
540 PRINT "AND YOU DID IT WITH "
TME" TIME UNITS LEFT"
550 Q = INT (TME * 1000 / SHOTS)

560 PRINT "-----> YOUR RATING
IS "Q
570 IF Q > HSCRE THEN HSCRE = Q
580 PRINT "-----> HIGHEST SCORE
SO FAR IS "HSCRE
600 PRINT : PRINT "DO YOU WANT A
NOTHER GAME (Y/N)? ";
610 GET T$: PRINT T$
620 IF T$ < > "Y" AND T$ < > "
N" THEN 610
630 IF T$ = "Y" THEN 40
640 PRINT "-----> OVER AND OUT
, CAPTAIN!"

```



```

650 END
660 REM *****
670 REM PLAYER MOVE
680 PRINT "WHICH SECTOR WILL YOU
      SHOOT AT?"
690 PRINT SPC( 8);"ACROSS? ";
700 GET S: PRINT S
710 IF S < 2 OR S > 9 THEN 700
720 PRINT SPC( 8);"AND DOWN? ";

730 GET R: PRINT R
740 IF R < 2 OR R > 9 THEN 730
750 IF A(R,S) = 1 THEN 170: REM
      ALIEN HIT
760 IF A(R,S) = 2 THEN PRINT "T
      HAT SECTOR IS ALREADY DESTRO
      YED!": FOR D = 1 TO 1000: NEXT
      D: RETURN
770 A(R,S) = 2
780 RETURN
790 REM *****
800 REM PRINT OUT
810 HOME
820 PRINT : PRINT "-> HIGHEST SC
      ORE SO FAR IS "HSCORE" <-"
830 PRINT "-> ALIEN SENSES DANGE
      R FACTOR " INT (100 * H / 3)
      / 100" <-"
840 PRINT "-----"
      "
850 PRINT "----> ALIEN NOW AT "N
      ;"/";M
860 PRINT "-> TIME LEFT: "TME"--
      - LASERS FIRED: "SHOTS
870 PRINT "-----"
      "
880 PRINT SPC( 5);"12345678910"

890 FOR K = 1 TO 10
900 PRINT " ";: IF K < 10 THEN
      PRINT " ";
910 PRINT K;
920 FOR J = 1 TO 10
930 IF K < 2 OR K > 9 OR J < 2 OR
      J > 9 THEN PRINT "-";: GOTO
      970
940 IF A(K,J) = 0 THEN PRINT "*"
      ";
950 IF A(K,J) = 1 THEN PRINT "A
      ";
960 IF A(K,J) = 2 THEN PRINT "
      ";
970 NEXT J
980 PRINT
990 NEXT K
1000 PRINT
1010 RETURN
1020 REM *****
1030 REM INITIALIZE
1040 HOME
1050 PRINT : PRINT : HTAB 16: PRINT
      "ENGULF"
1060 PRINT : PRINT "PLEASE STAND
      BY FOR YOUR MISSION..."
1070 TME = 30
1080 SHOTS = 0
1090 H = 0
1100 FOR B = 1 TO 10
1110 FOR C = 1 TO 10
1120 A(B,C) = 0
1130 IF B < 2 OR B > 9 OR C < 2 OR
      C > 9 THEN A(B,C) = 2
1140 NEXT C
1150 NEXT B
1160 M = INT ( RND (1) * 7) + 2
1170 N = INT ( RND (1) * 7) + 2
1180 A(M,N) = 1
1190 RETURN

```

MINE SWEEPER

There's trouble on the high seas and you're the only hope to clear a path through the heavily mined Juicy Straits Canal. Oil tankers must pass through safely with the mine-sweeper you're commanding controlling their fate.

Each section of the canal is split into a ten by fifteen grid and any location can contain a mine. Although you can't actually see the mines, your rusty old mine-detector can tell when a mine is in one of the eight locations that surround your position.



Where your ship is on the grid is shown by a number that also indicates how many mines are in the "square" of eight locations surrounding you. By backtracking, and by going around mines when you sense them, you can create a clear path from the left-hand side of the screen to the row of #'s on the right-hand side.

The only thing standing in your way are the mines. Land on one and it's all over. Move around the screen by entering U(p), D(own), R(ight), or L(ef). Be sure CAPS LOCK is on. You can exit at any time by pressing CONTROL-RESET. With skill and logic you'll be able to leave a clear trail of blank locations behind you for the tankers to use.




MINE SWEEPER was written by Neal Cavalier-Smith and has multiple levels of play. At the end of each round, the grid will be reprinted, with the mines shown in position. Your score is dependent upon how direct your path across the screen has been, and on the number of mines in that particular segment of the canal. Each time you finish a level, you'll be given a new grid, with more mines hidden.

Reading MINE SWEEPER

The following data strip contains the program described in this article. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, run the program from the Cauzin menu or enter RUN MINESWEEPER. Operating instructions are found in the article. Exit anytime by pressing CONTROL-RESET.



MINE SWEEPER
by Neal Cavalier-Smith
Second Giant Book of Computer Games
Ballantine Books
Copyright 1985. All rights reserved.

MINE SWEEPER

```

10 REM MINESWEEPER
11 REM
12 REM FROM TIM HARTNELL'S
13 REM SECOND GIANT BOOK
14 REM OF COMPUTER GAMES
15 REM
16 REM BALLANTINE BOOKS
17 REM
18 REM *****
19 REM
20 REM NEAL CAVALIER-SMITH
30 HOME
40 VTAB 10: HTAB 14: PRINT "MINE
  SWEEPER"
50 DIM A(11,17)
60 L = 1:S = 0:M = 0
70 REM *****
80 REM LAY MINES
90 FOR X = 1 TO 10
100 FOR Y = 1 TO 15
110 K = INT ( RND (1) * 25) + 1 -
  L
120 A(X,Y) = 46
130 IF K = 1 THEN A(X,Y) = 64
140 NEXT Y
150 A(X,1) = 46
160 NEXT X
170 C = 5:B = 1
180 A(C,B) = 33
190 GOTO 500
200 REM *****
210 REM PRINT MINEFIELD
220 HOME
230 PRINT : PRINT
240 FOR X = 1 TO 10
250 PRINT SPC( 11);">";
260 FOR Y = 1 TO 15
270 IF A(X,Y) < > 64 THEN PRINT
  CHR$( A(X,Y));
280 IF A(X,Y) = 64 THEN PRINT "
  .";
290 NEXT Y
300 PRINT "#"
310 NEXT X
320 PRINT
330 A(C,B) = 32
340 REM *****
350 REM MOVE SWEEPER
360 PRINT "ENTER YOUR MOVE U(P),
  D(OWN)"
370 PRINT ",R(IGHT), L(EFT) "
380 PRINT : PRINT : HTAB 20: GET
  B$: PRINT B$
390 C1 = 0:B1 = 0
400 IF B$ = "U" THEN C1 = - 1
410 IF B$ = "D" THEN C1 = 1
420 IF B$ = "R" THEN B1 = 1
430 IF B$ = "L" THEN B1 = - 1
440 IF C + C1 > 10 OR C + C1 < 1
  THEN 380
445 IF B + B1 < 1 THEN B1 = 0
450 IF B + B1 > 15 THEN 600
460 C = C + C1:B = B + B1
470 IF A(C,B) = 64 THEN 800
480 REM *****
490 REM COUNT ADJACENT MINES
500 N = 0
510 FOR K = - 1 TO 1
520 FOR D = - 1 TO 1
530 IF A(C + K,B + D) = 64 THEN
  N = N + 1
540 NEXT D
550 NEXT K
560 A(C,B) = 48 + N
570 M = M + 1
580 GOTO 210
590 REM *****
600 REM NEXT LEVEL ROUTINE
610 HOME
620 L = L + 1
630 PRINT "CONGRATULATIONS..."
640 PRINT "YOU HAVE CLEARED A PA
  TH THROUGH": PRINT
650 FOR X = 1 TO 10: PRINT SPC(
  11);"#";
660 FOR Y = 1 TO 15
670 PRINT CHR$( A(X,Y));
680 NEXT Y
690 PRINT "#"
700 NEXT X
710 PRINT : IF A(C,B) = 64 THEN
  END
720 PRINT "IT TOOK YOU "M" MOVES
  "
730 S = S + 100 - M
740 PRINT "YOUR SCORE IS "S
750 PRINT "YOU CAN NOW PROGRESS
  TO LEVEL "L
760 M = 0
770 PRINT "PRESS <RETURN> TO CON
  TINUE ";: GET Q$: PRINT
780 GOTO 80
790 REM *****
800 REM SPLAT
810 HOME : PRINT "SPLAT!!!!"
820 PRINT : PRINT "YOU'VE HIT A
  MINE, TURKEY, SO THE"
830 PRINT "GAME IS OVER. YOUR SC
  ORE IS "S + L * 10 - M
840 GOTO 650

```


IT'S A MAD, MAD WORLD

You were a fool to escape from the lunatic asylum. After all, life was good there, with made-to-measure straight-jackets and congenial company. Now you're in the big, wide world and it doesn't take long to discover that life can be pretty tough on the outside.

IT'S A MAD, MAD WORLD is a text-adventure program that will immediately put you into a number of odd situations. In fact, there are enough strange things happening to make you even more unbalanced than you already are.

Can you solve the problems and get back to the security and comfort of your padded cell? Only time will tell. You will always be faced with two choices at each decision point in the game. Exit the program early by pressing CONTROL-RESET. What you select will determine what happens in this flight of fancy and fun.






Reading IT'S A MAD, MAD WORLD

The following data strip contains the program described in this article. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, run the program from the Cauzin menu or enter **RUN MAD.WORLD**. Operating instructions are found in the article. Exit anytime by pressing **CONTROL-RESET**.



IT'S A MAD, MAD WORLD
by Tim Hartnell
Second Giant Book of Computer Games
Ballantine Books
Copyright 1985. All rights reserved.

IT'S A MAD, MAD WORLD

```

10 REM IT'S A MAD, MAD WORLD
11 REM
12 REM FROM TIM HARTNELL'S
13 REM SECOND GIANT BOOK
14 REM OF COMPUTER GAMES
15 REM
16 REM BALLANTINE BOOKS
17 REM
18 REM *****
19 REM
20 GOSUB 380: REM INITIALIZE
30 L = 1
40 HOME
50 PRINT SPC( 6);"IT'S A MAD, M
  AD WORLD"
60 PRINT : PRINT : PRINT " ";W$(
  L)
70 REM *****
80 PRINT
90 IF L = 21 THEN L = 20
100 IF W(L) < > 1 THEN 170
110 PRINT SPC( 2);"YOU WERE NOT
  CRAZY ENOUGH TO WIN THIS"
120 PRINT "ADVENTURE. IF YOU WAN
  T SOME ADVICE, TRY"
130 PRINT SPC( 3);"WRITING ADVE
  NTURE GAMES..."
140 PRINT : PRINT SPC( 6);"...T
  HAT WILL SEND YOU CRAZY!"
150 END
160 REM *****
170 IF W(L) < > 2 THEN 220
180 PRINT "WELL DONE. YOU MUST B
  E MAD TO FINISH"
190 PRINT SPC( 9);"THIS ADVENTU
  RE!": PRINT
200 END
210 REM *****
220 PRINT "DO YOU:"
230 PRINT
240 PRINT "1. ";A$(L)
250 PRINT
260 PRINT "2. ";B$(L)
270 PRINT : PRINT
280 HTAB 20: GET N$: PRINT N$
290 IF N$ = "" THEN 280
300 IF N$ = "1" THEN L = A(L): GOTO
  40
310 IF N$ = "2" THEN L = B(L): GOTO
  40
320 PRINT
330 PRINT SPC( 8);"NO SUCH CHOI
  CE, FOOL!"
340 GOTO 270
350 END
360 REM *****
370 REM INITIALIZE
380 HOME
390 DIM W$(25),A$(25),B$(25)
400 DIM W(25),A(25),B(25)
410 FOR X = 1 TO 25
420 READ W$(X),W(X)
430 READ A$(X),A(X)
440 READ B$(X),B(X)
450 NEXT X
460 RETURN
470 REM *****
480 REM DATA
490 DATA YOU HAVE CRASHED ON AN
  ISLAND,0
500 DATA HOBBLE ALONG THE BEACH,
  2,STAY BY THE HANGLIDER,10
510 DATA A ROARING TIGER APPEARS
  ,0
520 DATA RUN INTO THE FOREST,3,J
  UMP INTO THE WATER,11
530 DATA YOU SEE A DARK CAVE,0
540 DATA ENTER IT,4,KEEP GOING I
  NLAND,7
550 DATA YOU SEE A BAG OF GOLD,0

560 DATA LEAVE IT,5,TAKE IT,6
570 DATA A TROLL COMES BY AND TA
  KES THE GOLD. YOUR CHANCE
  TO BE RICH GOES!
580 DATA 0,CRY FOR HELP,10,PRACT
  ICE BEING A LAWNMOWER,25
590 DATA "A MONSTER APPEARS, YOU
  CANNOT RUN AWAY AS THE GOL
  D IS TOO HEAVY..."
600 DATA 1,0,0,0,0
610 DATA YOU SEE A BOTTLE OF WIN
  E,0
620 DATA DRINK IT,8,RUB THE BOTT
  LE,9
630 DATA "YOU FEEL RATHER HAPPY,
  PINK DISK DRIVES DANCE BEFO
  RE YOUR EYES"

```

640 DATA 1,0,0,0,0
650 DATA A GENIE APPEARS AND GIV
ES YOU A WISH,0
660 DATA WISH TO GO HOME,25,WISH
TO BE FILTHY RICH,5
670 DATA A SPACE CRAFT LANDS,0
680 DATA ENTER IT,12,JUMP IN THE
WATER,11
690 DATA YOU SEE A PIRATE SHIP,0

700 DATA GO TO SHORE,3,BOARD THE
SHIP,15
710 DATA THE ALIENS WANT TO CONQ
UER EARTH,0
720 DATA HELP THEM,14,FIGHT THEM
,13
730 DATA FOOLISH PERSON. YOU ARE
KILLED IN THE CONFLICT,1,
0,0,0,0
740 DATA A HARD BATTLE BUT YOU W
ON. YOU ARE STANDING IN
THE BATTLE FIELD
750 DATA 0,SURVEY THE BATTLEFIEL
D,16,REST AND DO NOTHING,14
760 DATA YOU BECOME A PIRATE FOR
YEARS BEFORE YOU ARE CAS
T ADRIPT IN A BATH
770 DATA 0,DO YOU WISH TO PRAY,9
,PADDLE FOR SHORE,20
780 DATA A WOUNDED HUMAN LIMPS T
OWARDS YOU,0
790 DATA KILL HIM,18,DRESS HIS W
OUNDS,17
800 DATA HE TURNS OUT TO BE THE
KING OF THE COUNTRY YOU
DEFEATED,0
810 DATA FLEE FROM HIM,3,ASK FOR
HIS FORGIVENESS,19
820 DATA "MISTAKE IN MESSAGE! S
HOULD HAVE READ 'HUMANOID',
AN ALIEN ALLY"
830 DATA 1,0,0,0,0
840 DATA YOU GET HIS FORGIVENESS
AND HE SENDS YOU BACK TO YO
UR HOME,2,0,0,0,0
850 DATA YOU ARE IN A BATH. THER
E IS A RUBBER DUCK AND A
SPONGE WITH YOU
860 DATA 0,STAY IN THE WARM BATH
,21,GET OUT AND DRY YOURSELF
,22
870 DATA I DON'T BLAME YOU AT AL
L...BUT YOU MUST DO SOMETHIN
G,20,0,0,0,0
880 DATA YOU'RE OUT AND DRY. DO
YOU WANT TO TAKE THE DUCKIE
WITH YOU?
890 DATA 0,OH YES PLEASE,23,NO T
HANK YOU,24
900 DATA CARRYING THE SACRED DUC
K IS PUNISHABLE BY DEATH HE
RE!,1,0,0,0,0
910 DATA YOU ARE IN A CLEARING I
N A FOREST,0
920 DATA "GO TO THE EAST, BACK T
O THE CAVE",3
930 DATA MAKE FOR THE PORT TO TH
E WEST,11
940 DATA YOU'RE BACK AT HOME IN
YOUR OWN PADDED CELL; THE J
ACKET FITS SNUGLY
950 DATA 2,0,0,0,0

THE CELTIC TAROT

Aren't computers wonderful! After saving the world from aliens, and fighting with a dragon or two, you can combine the world's newest technology with some of its elder wisdom, to find out what's in the cards for you.

Many experts believe that Tarot cards — widely used in fortune telling — appeared at the end of the 1300's in medieval France. "Whether the cards came from India, Egypt, or China... is of little moment," writes Frank Lind (*HOW TO UNDERSTAND THE TAROT*, The Aquarian Press, Wellingborough, UK, 1979). "What is of far more importance... is the antiquity of much of their symbolism" (pp. 7-8). The standard playing cards we use for games evolved from the original Tarot decks.

The 78 cards in the Tarot are divided into the Major and Minor Arcana. Fifty-six of the cards form four suits (Cups, Swords, Pentacles, and Wands). The 14th card in each suit is an additional court card, the knight, which was dropped when the ordinary deck of cards was formed. The 56 cards of the Minor Arcana have their own esoteric meanings and are often used in divination.



However, it is the Major Arcana (sometimes known as the "greater trumps") that is the real key to the Tarot. The CELTIC TAROT program concentrates on the 22 cards in this arcana. Occultists who use the Tarot claim the Major Arcana is like a mirror held up to your life, showing your grief and happiness, your plans and their fulfillment or destruction, your friendships, loves, and hatreds. Each card is many-sided, like a well-cut diamond, and each facet holds an insight.

There is no need for you to learn what each card signifies in order to use the program. Nor do you have to buy a Tarot deck, although you may well want to after running this program a few times.

There are many methods of using and interpreting the Tarot. Our program uses a pattern known as the "Celtic Cross." The Celtic design was chosen because it is "probably the most useful and versatile of all tarot spreads" (according to *PREDICTING THE FUTURE*, The Diagram Group, Ballantine Books, New York, 1983; p. 56).


The program does most of the work, leaving the finer points of interpretation to you. All you have to do is type in a question. The program deals your cards and reports the result. To exit early, press CONTROL-RESET.

Whether you feel that the Tarot is a path to knowledge or just a recreation, try the CELTIC TAROT.

Reading CELTIC TAROT

The following data strip contains the program described in this article. If you need additional help reading a data strip, refer to your reader instruction booklet. Your Cauzin communications program also contains help screens to assist you.

After you've read in the strip, run the program from the Cauzin menu or enter RUN CELTIC.TAROT. Operating instructions are found in the article. Exit anytime by pressing CONTROL-RESET.



CELTIC TAROT
by Tim Hartnell
Second Giant Book of Computer Games
Ballantine Books
Copyright 1985. All rights reserved.

CELTIC TAROT

```

10 REM THE CELTIC TAROT
11 REM
12 REM SECOND GIANT BOOK
13 REM OF COMPUTER GAMES
14 REM
15 REM BY TIM HARTNELL
16 REM
17 REM BALLANTINE BOOKS
18 REM
20 GOSUB 380: REM INITIALISE
30 GOSUB 250: REM ENTER QUESTIO
  N
40 GOSUB 530
50 PRINT R$;"?": PRINT
60 PRINT TAB( 6) "... LET ME SEE
  ...
70 GOSUB 530: GOSUB 530
80 REM *****
90 REM PREDICT
100 FOR J = 1 TO 10
110 GOSUB 510: REM DELAY/SPACE
  OUT
120 PRINT "*****
  *****"
130 PRINT A$(J)
140 GOSUB 530
150 X = INT ( RND (1) * 22): IF
  H(X) = 1 THEN 150
160 H(X) = 1
170 PRINT TAB( 13) "-----"

180 PRINT B$(X)
190 PRINT "*****
  *****"
200 GOSUB 530
210 NEXT J
220 GOSUB 510
230 END
240 REM *****
250 REM ENTER QUESTION
260 HOME
270 PRINT "ENTER YOUR QUESTION O
  R CONCERN NOW"
280 PRINT "AND I WILL USE THE CE
  LTIC CROSS SPREAD"
290 PRINT "IN AN ATTEMPT TO GIVE
  SOME GUIDANCE..."
300 GOSUB 510
310 INPUT "QUESTION: ";R$
320 GOSUB 510
330 PRINT TAB( 4) "PLEASE STAND
  BY..."
340 GOSUB 530
350 HOME
360 RETURN
370 REM *****
380 REM INITIALISE
390 HOME
395 VTAB 5: HTAB 12: PRINT "THE
  CELTIC TAROT"
400 REM ARRAYS
410 DIM A$(10),B$(21),H(21)
420 FOR J = 1 TO 10
430 READ A$(J)
440 NEXT J
450 FOR J = 0 TO 21
460 H(J) = 0
470 READ B$(J)
480 NEXT J
484 POKE - 16368,0: REM CLEAR K
  EYBOARD
485 VTAB 15: HTAB 9: PRINT "PRES
  S ANY KEY TO BEGIN. ";
486 X = PEEK ( - 16384): IF X <
  127 THEN X = RND (1): GOTO
  486: REM SCRAMBLE RANDOM NUM
  BERS
487 POKE - 16368,0
490 RETURN
500 REM *****
510 REM DELAY/SPACE OUT
520 PRINT : PRINT
530 FOR Z = 1 TO 1500: NEXT Z
540 RETURN
550 REM *****
560 REM INTRODUCTORY DATA
570 DATA "THIS IS WHERE YOU STA
  ND NOW..."
580 DATA "THIS IS THE NEAR FUTU
  RE..."
590 DATA "THIS IS WHAT YOU MIGH
  T DO..."
600 DATA "THIS IS SOMETHING IN
  THE PAST THAT IS RELEVANT
  TO YOUR QUESTION..."
610 DATA "THIS CARD RELATES TO
  A MORE RECENT PAST EVENT..."

```

620 DATA "NOW, THIS CARD INDICA
 TES YOUR POSITION IN SIX MO
 NTHS OR SO..."
 630 DATA "IN YOUR LIFE OR WORK,
 THIS MOST CLOSELY INFLUENCE
 S YOU..."
 640 DATA "THIS CARD REPRESENTS
 SOCIAL INFLUENCES ON YOUR Q
 UERY..."
 650 DATA "YOUR HOPES AND FEARS
 ARE REFLECTED IN THIS CARD
 ..."
 660 DATA "AND, FROM THESE, IN R
 EPLY, THIS CARD POINTS TO
 WARD AN ANSWER..."
 670 REM *****
 680 REM CARD DATA
 690 DATA "THE FOOL: NEW BEGINNI
 NGS AND CHALLENGES"
 700 DATA "THE MAGICIAN: OUTWARD
 PROGRESS, DECISIONS
 TO BE MADE..."
 710 DATA "THE HIGH PRIESTESS: W
 ISDOM, A PROBLEM WILL BE S
 IMPLIFIED..."
 720 DATA "THE EMPRESS: A STABLE
 SITUATION, GOOD FORTUNE..
 ."
 730 DATA "THE EMPEROR: GOALS WI
 LL BE REACHED, BEWARE OF
 RUTHLESSNESS..."
 740 DATA "THE HIGH PRIEST: LIST
 EN TO ADVICE, UNDERSTAN
 DING LIKELY..."
 750 DATA "THE LOVERS: A POSITIV
 E DECISION LIKELY, RELATIONS
 HIPS PROSPER..."
 760 DATA "THE CHARIOT: A GOOD
 STAGE IN YOUR LIFE ATTAINED
 , TRAVEL LIKELY..."
 770 DATA "JUSTICE: BALANCE WILL
 BE REACHED, DUE REWAR
 D LIKELY..."
 780 DATA "THE HERMIT: REVALUATI
 ON OF PRIORITIES INDICATED
 , INNER GROWTH"
 790 DATA "WHEEL OF FORTUNE: GRO
 WTH, LUCK, DECISIONS
 TO BE MADE..."
 800 DATA "STRENGTH: YOUR INNER
 QUALITIES WILL ENSURE TH
 AT YOU MEET THE TEST..."
 810 DATA "HANGED MAN: CAST AWAY
 MATERIAL VALUES, SACRIFICE
 REWARDED..."
 820 DATA "DEATH: FRESH AWARENES
 S WILL COME FROM TEMPORARY
 SETBACKS..."
 830 DATA "ANGEL OF TIME (TEMPER
 ANCE): MODERATION DEMANDED;
 WISE CHOICE SEEN"
 840 DATA "THE DEVIL: BE WARY OF
 GIVING IN TOO EASILY TO
 YOUR WHIMS"
 850 DATA "THE TOWER: UNFORTUNAT
 E INDICATIONS, REGROWTH
 CAN BE EXPECTED..."
 860 DATA "THE STAR: VERY POSITI
 VE, FRESH DEVELOPME
 NTS SEEN..."
 870 DATA "THE MOON: TRUST YOUR
 HEART RATHER THAN YOUR HEAD
 ..."
 880 DATA "THE SUN: A GOAL WITHI
 N REACH, LIGHT AFTER DAR
 KNESS..."
 890 DATA "JUDGEMENT: A PERIOD O
 F FRESH BEGINNINGS AND ASSES
 SING VALUES..."
 900 DATA "THE WORLD: COMPLETION
 OF A DIFFICULT TASK, ATT
 AINMENT OF A GOAL"

CAUZIN'S CORNER...

and now for something slightly different

These six games are written in BASIC, so you can see how they work and modify them. To see the commands, LOAD the program and enter LIST. You'll see all the program lines scroll down the screen. Enter LIST 200 to see just one line, in this case line 200.

If you want a printout of the program, enter PR#1 and then enter LIST. It will be helpful to have a printout to help you learn more about BASIC and about these games.

Many moves are determined by random chance. In BASIC this is the RND(1) statement. To improve your chances of winning or to make the game harder, look for the use of this command and change the odds.

For example, ENGULF uses this command in two lines to determine a random move for the alien.

```
370 M=M - INT(RND(1)*3) + INT(RND(1)*3)
410 N=N - INT(RND(1)*3) + INT(RND(1)*3)
```

Variables M and N are the grid location of the alien. These two lines randomly change the coordinates anywhere from -2 to +2. To limit the alien's moves, change all these three's to two's. To increase the alien's range change some of the three's to 3.5 or 4.

You can easily add a printout function to CELTIC TAROT. Add these two lines to get a printout of what the cards say in response to your question.

```
45 PRINT CHR$(4);"PR#1":PRINT
225 PRINT CHR$(4);"PR#0"
```

Line 45 sends all output to your printer. Output is returned to the screen by line 225.

These games all use the text screen. You could study them and use the programs as algorithms, or models, and add your own graphics and sounds. This can be as simple or complicated as you care to make it.

For example, try using "text graphics" to add color to CAVALIER 'CHUTE. These are graphics created with a PRINT statement while in lo-res (GR) mode. After you LOAD the program, type in the following lines. They will automatically take the place of the previous ones. Save the modified program under a new name, to preserve the original version.

```
60 GR: GOSUB 360: REM PRINT CHUTE
380 VTAB STP + 1
390 REM
400 REM
410 HTAB ACC + 1: PRINT "/"
420 HTAB ACC: PRINT "///"
430 HTAB ACC + 1: PRINT ";"
440 HTAB ACC + 1: PRINT ";"
550 HTAB PAD: PRINT "nnnn"
600 HTAB PAD + 2: GET Y$
```

LIMITED WARRANTY

Cauzin StripWare™

835 South Main Street
Waterbury, Connecticut 06706

CAUZIN SYSTEMS, INC. (CAUZIN) warrants this package against defects in materials and workmanship for a period of 90 days from the date of original purchase.

If you discover a defect, CAUZIN will, at its option, either repair, replace, or refund the purchase price of this product at no charge to you, provided it is returned during the warranty period indicated above or any extended period(s) paid for by the user. Transportation charges should be prepaid, to the authorized dealer from whom you purchased it or to any other authorized CAUZIN dealer in the country.

You may obtain additional information from CAUZIN directly at the address printed on this certificate, or by calling (203) 573-0150. Please attach your name, address, telephone number, a description of the problem, and a copy of a bill of sale as proof of date of original purchase, to each product returned for warranty service.

This warranty applies only to products manufactured by or for CAUZIN, which can be identified by the CAUZIN

trademark, trade name, or logo affixed to the product. CAUZIN does not warrant any product(s) that are not CAUZIN products. This warranty does not apply if the product has been damaged by accident, abuse, misuse or misapplication, has been modified without the written permission of CAUZIN, or if any CAUZIN serial number has been removed, altered, or defaced.

ALL WARRANTIES, EXPRESSED OR IMPLIED, INCLUDING WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE, ARE LIMITED IN DURATION TO 90 DAYS FROM THE ORIGINAL DATE OF PURCHASE. THERE ARE NO WARRANTIES WHICH EXTEND BEYOND THE DESCRIPTION ON THE FACE HEREOF.


THE WARRANTY AND REMEDIES SET FORTH ABOVE ARE EXCLUSIVE AND IN LIEU OF ALL OTHERS, ORAL OR WRITTEN, EXPRESSED OR IMPLIED. NO CAUZIN DEALER, AGENT, OR EMPLOYEE IS AUTHORIZED TO MAKE ANY MODIFICATION, OR ADDITION TO THIS

WARRANTY.

CAUZIN IS NOT RESPONSIBLE FOR SPECIAL, INCIDENTAL, OR CONSEQUENTIAL DAMAGES RESULTING FROM ANY BREACH OF WARRANTY, OR UNDER ANY OTHER LEGAL THEORY, INCLUDING LOST PROFITS, DOWNTIME, GOODWILL, DAMAGE TO, OR REPLACEMENT OF EQUIPMENT AND PROPERTY, AND ANY COST OF RECOVERING, REPROGRAMMING OR REPRODUCING ANY PROGRAM OR DATA STORED IN OR USED WITH CAUZIN PRODUCTS.

Some states do not allow the exclusion or limitation of incidental or consequential damages or limitations on how long an implied warranty lasts, so the above limitations or exclusion may not apply to you.

This warranty shall not be applicable to the extent that any provision of this warranty is prohibited by an federal, state or municipal law which cannot be preempted. This warranty gives you specific legal rights, and you may also have other rights which vary from state to state.



GET ORGANIZED!

Order your Cauzin Softstrip™ System high quality binder today! Designed especially to hold Cauzin StripWare™ packages, Communication program disk, and your data disk.

This attractive binder will help you keep track of your StripWare, disks, and makes a lovely addition to your bookshelf.

Available only through:
Cauzin Systems, Inc.
835 Main Street
Waterbury, CT 06706

Send check or money order for \$12 to cover cost, postage and handling.

- Get out of tight places in
THE BIG MAZE
- Survive with subtle precision in
CAVALIER 'CHUTE
- ENGULF aliens in far away galaxies
- Explosive action abounds in
MINE SWEEPER
- IT'S A MAD, MAD WORLD full
of adventure
- THE CELTIC TAROT forsees what
lies ahead

CAUZIN SYSTEMS, INC.
835 South Main Street
Waterbury, Conn. 06706
(203) 573-0150

850.3005